Statistics 231B SAS Practice Lab #8
Spring 2006

This lab is designed to give the students practice in logistic regression model goodness of fit test, model diagnosis and prediction.
Example:  A marketing research firm was engaged by an automobile manufacturer to conduct a pilot study to examine the feasibility of using logistic regression for ascertaining the likelihood that a family will purchase a new car during the next year. A random sample of 33 suburban families was selected. Data on annual family income (X1, in thousand dollars) and the current age of the oldest family automobile (X2, in years) were obtained. A follow-up interview conducted 12 months later was used to determine whether the family actually purchased a new car (Y=1) or did not purchase a new car (Y=0) during the year. Data were recorded in the file CH14PR13.txt. 
1. Logistic regression model diagnosis

SAS CODE:

/*read in file ch14pr13.txt one by one and create a column case for /*case number obtain the centered x1 values*/ 

data ch14pr13;

infile 'c:\stat231B06\ch14pr13.txt';

do case=1 to 33;

input y x1 x2;

x1c=x1-38.2424;/*mean(X1)=38.2424*/
output;

end;

run;

/*In the output statement immediately following the model statement*/
/*we set yhat=p(estimated probability, ),chires=reschi(pearson */

/*residual, ), devres=resdev(deviance residual,devi),difchsq=difchisq*/

/*(delta chi-square  ), difdev=difdev(delta deviance,),*/

/*hatdiag=h(diagonal elements of the hat matrix*/
/*Here p,reschi,resdev,difchisq, difdev and h are SAS key words for*/

/*the respective variables. We then output them into results1 so that*/ /*the data for all variables will be assessable in one file. */
ods html; 

ods graphics on; /*ods graphics version of plots are produced*/
proc logistic data=ch14pr13;

model y (event='1')=x1c;

output out=results1 p=yhat reschi=chires resdev=devres difchisq=difchisq difdev=difdev h=hatdiag;

run;

/*use equation 14.81 (text page 592) to find rsp (studentized Pearson*/

/*residual, rspi) and use equation 14.87 (text page 600) to find */

/*cookd (COOK's Distance, Di)*/
data results2;

set results1;

rsp=chires/(sqrt(1-hatdiag));

cookd=((chires**2)*hatdiag)/(5*(1-hatdiag)**2);

proc gplot data=results2;/*SAS output 1*/
symbol1 color=black value=none interpol=join;

plot hatdiag*case;

run; 

proc gplot data=results2; /*SAS output 2*/
symbol1 color=black value=none interpol=join;

plot difchisq*case;

run;

proc gplot data=results2; /*SAS output 3*/
symbol1 color=black value=none interpol=join;

plot difdev*case;

run;

proc gplot data=results2; /*SAS output 4*/
symbol1 color=black value=none interpol=join;

plot cookd*case;

run;

ods graphics off;

ods html close;
a. For the logistic regression model obtained according to the SBC criterion, prepare an index plot of the diagonal elements of the estimated hat matrix (14.80) of the text ALSM. Use the plot to identify any outlying X observations.

The logistic regression model obtained according to the SBC criterion contains the predictor X1c. That is X1c=X1-38.2424.
SAS OUTPUT1:

The leverage plot identifies case 7 and 10 as being somewhat outlying in the X space- and therefore potentially influential

Note: To find out the case 7 and 10, you can also use SAS  proc sort procedure.

data results2;

proc sort data=results2;

by hatdiag;

run;

proc print;

var case hatdiag;

run;
SAS OUTPUT:
                                                                 Case   hatdiag

                                      32      7     0.13163

                                      33     10     0.13766
      b. To assess the influence of individual observations, obtain the delta chi-square statistic (14.85), the delta deviance statistic (14.86), and Cook’s distance (14.87) for each observation. Plot each of these in separate index plots and identify any influential observations. Summarize your findings.
SAS OUTPUT 2:

delta chi-square statistic (14.85) VS. case number

The index plots of the delta chi-square have 4 spikes corresponding to cases 9,19,26 and 29.
Note: To find out the four cases , you can also use SAS  proc sort procedure.

proc sort data=results2;

by difchisq;

run;

proc print;

var case difchisq;

run;
Partial SAS OUTPUT
                                                              Case   difchisq
                                     30      9      2.80785

                                     31     29      2.93290

                                     32     19      3.51236

                                     33     26      4.02128
SAS OUTPUT 3:

The index plots of the delta deviance have 4 spikes corresponding to cases 9,19,26 and 29.
Note: To find out the four cases , you can also use SAS  proc sort procedure.

proc sort data=results2;

by difdev;

run;

proc print;

var case difdev;

run;

Partial SAS OUTPUT
                                                               Case     difdev
                                      30      9     2.80039

                                      31     29     2.80826

                                      32     19     3.12016

                                      33     26     3.37066
SAS OUTPUT 4:

The plot of Cook’s distances indicates that cases 9,19,26,29 and 30 are 

most influential in terms of effect on the linear predictor.
Note: To find out the five cases , you can also use SAS  proc sort procedure.

proc sort data=results2;

by cookd;

run;

proc print;

var case cookd;

run;

Partial SAS OUTPUT
                                                            Case  cookd

                                     29     29     0.030336

                                     30     30     0.037383

                                     31     19     0.041238

                                     32     26     0.051529

                                     33      9     0.061642
3. Inferences about Mean Response
Pointer Estimator:

Denote the vector of the levels of the X variables for which ( is to be estimated by Xh: 

   The point estimator of (h will be denoted by 

where b is the vector of estimated regression coefficients.

Interval Estimation

Step 1: calculate confidence limits for the logit mean response 

Approximate 1-( large-sample confidence limits for the logit mean response 

Step 2: From the fact that 
[image: image11.wmf]1

'

)]

exp(

1

[

-

p

-

+

=

p

h

h

, the approximate 1-( large-sample confidence limits L* and U* for the mean response 

Simultaneous Confidence Intervals for Several Mean Responses

When it is desired to estimate several mean response (h corresponding to different Xh vectors with family confidence coefficient 1-(, Bonferroni simultaneous confidence intervals may be used. 

Step 1: The g simultaneous approximate 1-( large-sample confidence limits for the logit mean response 

Step 2: The g simultaneous approximate 1-( large-sample confidence limits for the mean response 

(a) For the disease outbreak data ch14ta03.txt. See below.

	Example:

On the basis of the fitted logistic regression function based on x1, x2,x3 and x4, obtain a 95 percent confidence interval for the mean response (h for a person 10 years old who are of lower socioeconomic status and live in sector 1 have contracted disease
SAS CODE:

data ch14ta03;

infile 'c:\stat231B06\ch14ta03.txt' DELIMITER='09'x;

input case x1 x2 x3 x4 y;

/*the outest= and covout options create a data set that contains parameter estimates*/

/*and their covariances for the model fitted*/ 
proc logistic data=ch14ta03 outest=betas covout;

model y (event='1')= x1 x2 x3 x4 /covb; 

run;

/*add column for zscore*/
data betas;

set betas;

zvalue=probit(0.975);

run;

proc iml;

use betas;

read all var {'Intercept'} into a1;

read all var {'x1'} into a2;

read all var {'x2'} into a3;

read all var {'x3'} into a4;

read all var {'x4'} into a5;

read all var {'zvalue'} into zvalue;
/*define the b vector*/
b=t(a1[1]||a2[1]||a3[1]||a4[1]||a5[1]);

/*define variance-covariance matrix for b*/
covb=(a1[2:6]||a2[2:6]||a3[2:6]||a4[2:6]||a5[2:6]);

/*Define xh matrix*/
xh={1,10,0,1,0};

/*find the point estimator of logit mean based on equation above the equation (14.91)*/

/*on page 602 of text ALSM*/
logitpi=xh`*b;

/*find the estimated variance of logitpi based on the equation (14.92)on page 602*/

/*of text ALSM*/
sbsqr=xh`*covb*xh;

/*find confidence limit for logit mean*/
L=logitpi-zvalue*sqrt(sbsqr);

U=logitpi+zvalue*sqrt(sbsqr);

L=L[1];

U=U[1];

/*find confidence limit L* and U* for probability based on equation on page 603*/

/*of text ALSM*/
Lstar=1/(1+EXP(-L));

Ustar=1/(1+EXP(-U));

print L U Lstar Ustar;   /*SAS output 5*/
quit;
SAS OUTPUT 5:
        L         U     LSTAR     USTAR

-3.383969 -1.256791 0.0328002 0.2215268
4. Prediction of a New Observation

1. Choice of Prediction Rule

       Where is the cutoff point?

(1) Use .5 as the cutoff 

  ((a) it is equally likely in the population of interest that  

                                                       outcomes 0 and 1 will occur; and 

                                                 (b) the costs of incorrectly predicting 0 and 1 are 

                                                       approximately the same

(2) Find the best cutoff for the data set on which the multiple logistic regression model is based. 

For each cutoff, the rule is employed on the n cases in the model-building data set and the proportion of cases incorrectly predicted is ascertained. The cutoff for which the proportion of incorrect predictions is lowest is the one to be employed.
( (a) the data set is a random sample from the relevant population, and thus reflects 
the proper proportions of 0s and 1s in the population

      (b) the costs of incorrectly predicting 0 and 1 are approximately the same

(3) Use prior probabilities and costs of incorrect predictions in determining the cutoff.

((a) prior information is available about the likelihood of 1s and 0s in the population

(b) the data set is not a random sample from the population

(c) the cost of incorrectly predicting outcome 1 differs substantially from the cost                 
    of incorrectly predicting outcome 0.

a. For the disease outbreak example, a prediction rule is to be based on the fitted regression function based on x1,x2,x3 and x4. For the sample cases, find the total error rate, the error rate for person contracting disease, and  the error rate for person not contracting disease for the following cutoff 0.316:


First rule: 

b. Obtain the area under the ROC curve to assess the model’s predictive power here. 

      What do you conclude?

SAS CODE:
/*save the predicted probability in a variable named yhat (p=yhat)*/ /*and then set newgp=1 if yhat is greater than or equal to 0.316*/

/*otherwise newgp=0. We then form a 2 by 2 table of Y values */

/*(disease versus nondiseased) by newgp values (predicted diseased */

/*versus predicted nondiseases)*/

/*the outroc option on the model statement saves the data necessary*/

/*to produce the ROC curve in a data file named roc (outroc=roc).*/

/*Two of the variables stored in roc data are _sensit_(sensitivity)*/

/*and _1mspec_(1-specificity). In the proc gplot we plot _sensit_ */

/*against _1mspec_ to form the ROC curve*/

proc logistic data=ch14ta03 descending;

model y=x1 x2 x3 x4/outroc=roc;

output out=results p=yhat;

run;

data results;

set results;

if yhat ge .316 then newgp=1; else newgp=0;

proc freq data=results; /*SAS OUTPUT 6*/
table y*newgp/ nopercent norow nocol;

run;

/*ROC curve*/
proc gplot data=roc;

symbol1 color=black value=none interpol=join;

plot _SENSIT_*_1MSPEC_;

run;
SAS OUTPUT 6:

The FREQ Procedure

                                      Table of y by newgp

                              y         newgp

                              Frequency‚       0‚       1‚  Total

                              ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                                     0 ‚     49 ‚     18 ‚     67

                              ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                                     1 ‚      8 ‚     23 ‚     31

                              ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                              Total          57       41       98

(a)
There were 8+23=31 (false negatives+true positives) diseased individuals and 49+19=67 (true negatives +false positives) non-diseased individuals. The fitted regression function correctly predicted 23 of the 31 diseased individuals. Thus the error rate for person contracting disease is 8/31=0.258. The fitted regression function correctly predicted 49 of the 67 non-diseased individuals. Thus the error rate for person not contracting disease is 18/67=0.2687. The total error rate is (18+8)/98=0.2653.

Use the same SAS code and by changing value 0.316 to other values, we can get the three error rate for a series of cutoffs and the cutoff minimizes the total error rate can be easily obtained. 
(b) 

                 Association of Predicted Probabilities and Observed Responses

                       Percent Concordant     77.5    Somers' D    0.554

                       Percent Discordant     22.1    Gamma        0.556

                       Percent Tied            0.3    Tau-a        0.242

                       Pairs                  2077    c            0.777
Note that the area under the ROC curve is given by the statistic c  in the “Association of Predicted Probabilities and Observed Responses” table. In this example, the area under the ROC curve is 0.777. This means that the predictions were better than random guess which would have the value 0.5
5. Validation of Prediction Error rate


a. How can you establish whether the observed total error rate for the best cutoff is a reliable indicator of the predictive ability of the fitted regression function and the chosen cutoff?

The reliability of the prediction error rate observed in the model-building data set is examined by applying the chosen prediction rule to a validation data set. If the new prediction error rate is about the same as that for the model-building data set, then the latter gives a reliable indication of the predictive ability of the fitted logistic regression model and the chosen prediction rule. If the new data lead to a considerably higher prediction error rate, then the fitted logistic regression model and the chosen prediction rule do not predict new observations as well as originally indicated.
	For the disease outbreak example, the fitted logistic regression function based on the model-building data set is

We use this fitted logistic regression function to calculate estimated probabilities 

for cases 99-196 in the disease outbreak data set in Appendix C.10. The chosen prediction rule is 

We want to classify observations in the validation data set.

data disease;                                                                                                                          

infile 'c:\stat231B06\appenc10.txt';                                                                                    

input case age status sector y other ;                                                                                                             

x1=age;

/*code the status with two dummy var*/

if status eq 2 then x2=1; else x2=0;

if status eq 3 then x3=1; else x3=0;

/*code sector with one dummy var*/

if sector eq 2 then x4=1; else x4=0;

/*calculate estimated probabilities*/

/*based on fitted logistic model*/

yhat=(1+exp(2.3129-(0.02975*x1)-(0.4088*x2)+(0.30525*x3)-(1.5747*x4)))**-1;

/*make group predictions based on */

/*the specified prediction rule*/

if yhat ge .325 then newgp=1; else newgp=0;

run;

/*generate the frequency table for */

/*the validation data set */

/*(cases 98-196)*/

proc freq;

where case>98;

table y*newgp/nocol nopct;

run;


           The FREQ Procedure

                                      Table of y by newgp

                              y         newgp

                              Frequency‚

                              Row Pct  ‚       0‚       1‚  Total

                              ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                                     0 ‚     44 ‚     28 ‚     72

                                       ‚  61.11 ‚  38.89 ‚

                              ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                                     1 ‚     12 ‚     14 ‚     26

                                       ‚  46.15 ‚  53.85 ‚

                              ƒƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆƒƒƒƒƒƒƒƒˆ

                              Total          56       42       98

The off-diagonal percentages indicate that 12(46.2%) of the 26 diseased individuals were incorrectly classified (prediction error) as non-diseased. Twenty-eight (38.9%) of the 72 non-diseased individuals were incorrectly classified as diseased. Note that the total prediction error rate of 40.8% =(28+12)/98 is considerably higher than the 26.5% error rate based on the model-building data set. Therefore the fitted logistic regression model and the chosen prediction rule do not predict new observations as well as originally indicated.






_1209799416.unknown

_1209800599.unknown

_1209892263.unknown

_1209979795.unknown

_1209979796.unknown

_1209979797.unknown

_1209979794.unknown

_1209892330.unknown

_1209802616.unknown

_1209844782.unknown

_1209844912.unknown

_1209802563.unknown

_1209799935.unknown

_1209800499.unknown

_1209800511.unknown

_1209800038.unknown

_1209800052.unknown

_1209799729.unknown

_1209799118.unknown

_1209799303.unknown

_1209799008.unknown

